

# 6

---

## Product Mix Problems

### 6.1 Introduction

Product mix problems are conceptually the easiest constrained optimization problems to comprehend. The Astro/Cosmo problem considered earlier is an example. Although product mix problems are seldom encountered in their simple textbook form in practice, they very frequently constitute important components of larger problems such as multiperiod planning models.

The features of a product mix problem are that there is a collection of products competing for a finite set of resources. If there are  $m$  resources and  $n$  products, then the so-called “technology” is characterized by a table with  $m$  rows and  $n$  columns of technologic coefficients. The coefficient in row  $i$ , column  $j$ , is the number of units of resource  $i$  used by each unit of product  $j$ . The numbers in a row of the table are simply the coefficients of a constraint in the LP. In simple product mix problems, these coefficients are nonnegative. Additionally, associated with each product is a profit contribution per unit and associated with each resource is an availability. The objective is to find how much to produce of each product (i.e., the mix) to maximize profits subject to not using more of each resource than is available.

The following product mix example will illustrate not only product mix LP formulations, but also: 1) representation of nonlinear profit functions and 2) the fact that most problems have alternative correct formulations. Two people may develop different formulations of the same problem, but both may be correct.

## 6.2 Example

A certain plant can manufacture five different products in any combination. Each product requires time on each of three machines in the following manner (figures in minutes/unit):

Product	Machine		
	1	2	3
<b>A</b>	12	8	5
<b>B</b>	7	9	10
<b>C</b>	8	4	7
<b>D</b>	10	0	3
<b>E</b>	7	11	2

Each machine is available 128 hours per week.

Products *A*, *B*, and *C* are purely competitive and any amounts made may be sold at respective prices of \$5, \$4, and \$5. The first 20 units of *D* and *E* produced per week can be sold at \$4 each, but all made in excess of 20 can only be sold at \$3 each. Variable labor costs are \$4 per hour for machines 1 and 2, while machine 3 labor costs \$3 per hour. Material costs are \$2 for products *A* and *C*, while products *B*, *D*, and *E* only cost \$1. You wish to maximize profit to the firm.

The principal complication is that the profit contributions of products *D* and *E* are not linear. You may find the following device useful for eliminating this complication. Define two additional products  $D_2$  and  $E_2$ , which sell for \$3 per unit. What upper limits must then be placed on the sale of the original products *D* and *E*? The decision variables and their profit contributions are as follows:

Decision Variables	Definition	Profit Contribution per Unit
<i>A</i>	Number of units of <i>A</i> produced per week	$5 - 2 = \$3$
<i>B</i>	Number of units of <i>B</i> produced per week	$4 - 1 = \$3$
<i>C</i>	Number of units of <i>C</i> produced per week	$5 - 2 = \$3$
<i>D</i>	Number of units of <i>D</i> not in excess of 20 produced/week	\$3
$D_2$	Number of units of <i>D</i> produced in excess of 20 per week*	\$2
<i>E</i>	Number of units of <i>E</i> not in excess of 20 produced/week	\$3
$E_2$	Number of units of <i>E</i> produced in excess of 20	\$2
$M_1$	Hours of machine 1 used per week	-\$4
$M_2$	Hours of machine 2 used per week	-\$4
$M_3$	Hours of machine 3 used per week	-\$3

\*Total production of product *D* is  $D + D_2$ .

We will not worry about issues of sequencing the various products on each machine. This is reasonable if the due-dates for the products are far enough in the future. Our problem in this case is to:

Maximize Revenues minus costs  
 Subject to  
 Minutes used equals minutes run on each machine,  
 At most 20 units each can be produced of products  $D$  and  $E$ ,  
 Each machine can be run at most 128 hours.

More precisely, the formulation in LINGO is:

```
! Maximize revenue minus costs;
MAX = 3 * A + 3 * B + 3 * C + 3 * D + 2 * D2 + 3 * E
      + 2 * E2 - 4 * M1 - 4 * M2 - 3 * M3;
! Machine time used = machine time made available;
12*A + 7*B + 8*C + 10*D + 10*D2 + 7*E + 7*E2 - 60*M1 = 0;
8*A + 9*B + 4*C + 11*E + 11*E2 - 60*M2 = 0;
5*A + 10*B + 7*C + 3*D + 3*D2 + 2*E + 2*E2 - 60*M3=0;
D <= 20; ! Max sellable at high price;
E <= 20;
!Machine availability;
M1 <= 128;
M2 <= 128;
M3 <= 128;
END
```

The first three constraints have the units of “minutes” and specify the hours of machine time as a function of the number of units produced. The next two constraints place upper limits on the number of high profit units of  $D$  and  $E$  that may be sold. The final three constraints put upper limits on the amount of machine time that may be used and have the units of “hours”.

Constraint 2 can be first written as:

$$\frac{12A + 7B + 8C + 10D + 10D_2 + 7E + 7E_2}{60} = M_1$$

Multiplying by 60 and bringing  $M_1$  to the left gives the second constraint. The solution is:

```
Optimal solution found at step: 4
Objective value: 1777.625
Variable Value Reduced Cost
A 0.0000000 1.358334
B 0.0000000 0.1854168
C 942.5000 0.0000000
D 0.0000000 0.1291668
D2 0.0000000 1.129167
E 20.00000 0.0000000
E2 0.0000000 0.9187501
M1 128.0000 0.0000000
M2 66.50000 0.0000000
M3 110.6250 0.0000000
```

Row	Slack or Surplus	Dual Price
1	1777.625	1.000000
2	0.000000	0.2979167
3	0.000000	0.6666667E-01
4	0.000000	0.500000E-01
5	20.00000	0.000000
6	0.000000	0.812500E-01
7	0.000000	13.87500
8	61.50000	0.000000
9	17.37500	0.000000

The form of the solution is quite simple to state: make as many of  $E$  as possible (20). After that, make as much of product  $C$  as possible until we run out of capacity on machine 1.

This problem is a good example of one for which it is very easy to develop alternative formulations of the same problem. These alternative formulations are all correct, but may have more or less constraints and variables. For example, the constraint:

$$8A + 9B + 4C + 11E + 11E_2 - 60M_2 = 0$$

can be rewritten as:

$$M_2 = (8A + 9B + 4C + 11E + 11E_2)/60.$$

The expression on the right-hand side can be substituted for  $M_2$  wherever  $M_2$  appears in the formulation. Because the expression on the right-hand side will always be nonnegative, the nonnegativity constraint on  $M_2$  will automatically be satisfied. Thus,  $M_2$  and the above constraint can be eliminated from the problem if we are willing to do a bit of arithmetic. When similar arguments are applied to  $M_1$  and  $M_3$  and the implied divisions are performed, one obtains the formulation:

```

MAX = 1.416667*A + 1.433333*B + 1.85*C + 2.183334*D + 1.183333*D2 +
1.7*E + .7*E2;
! Machine time used = machine time made available;
12*A + 7*B + 8*C + 10*D + 10*D2 + 7*E + 7*E2 <= 7680;
8*A + 9*B + 4*C + 11*E + 11*E2 <= 7680;
5*A + 10*B + 7*C + 3*D + 3*D2 + 2*E + 2*E2 <= 7680;
! Product limits;
D < 20;
E < 20;

```

This looks more like a standard product mix formulation. All the constraints are capacity constraints of some sort. Notice the solution to this formulation is really the same as the previous formulation:

Optimal solution found at step:	6	
Objective value:	1777.625	
Variable	Value	Reduced Cost
A	0.0000000	1.358333
B	0.0000000	0.1854170
C	942.5000	0.0000000
D	0.0000000	0.1291660
D2	0.0000000	1.129167
E	20.00000	0.0000000
E2	0.0000000	0.9187500
Row	Slack or Surplus	Dual Price
1	1777.625	1.000000
2	0.0000000	0.2312500
3	3690.000	0.0000000
4	1042.500	0.0000000
5	20.00000	0.0000000
6	0.0000000	0.8125000E-01

The lazy formulator might give the first formulation, whereas the second formulation might be given by the person who enjoys doing arithmetic.

### 6.3 Process Selection Product Mix Problems

A not uncommon feature of product mix models is two or more distinct variables in the LP formulation may actually correspond to alternate methods for producing the same product. In this case, the LP is being used not only to discover how much should be produced of a product, but also to select the best process for producing each product.

A second feature that usually appears with product mix problems is a requirement that a certain amount of a product be produced. This condition takes the problem out of the realm of simple product mix. Nevertheless, let us consider a problem with the above two features.

The American Metal Fabricating Company (AMFC) produces various products from steel bars. One of the initial steps is a shaping operation performed by rolling machines. There are three machines available for this purpose, the  $B_3$ ,  $B_4$ , and  $B_5$ . The following table gives their features:

Machine	Speed in Feet per Minute	Allowable Raw Material Thickness in Inches	Available Hours per Week	Labor Cost Per Hour Operating
$B_3$	150	3/16 to 3/8	35	\$10
$B_4$	100	5/16 to 1/2	35	\$15
$B_5$	75	3/8 to 3/4	35	\$17

This kind of combination of capabilities is not uncommon. That is, machines that process larger material operate at slower speed.

This week, three products must be produced. AMFC must produce at least 218,000 feet of  $\frac{1}{4}$ " material, 114,000 feet of  $\frac{3}{8}$ " material, and 111,000 feet of  $\frac{1}{2}$ " material. The profit contributions per foot excluding labor for these three products are 0.017, 0.019, and 0.02. These prices apply to all

production (e.g., any in excess of the required production). The shipping department has a capacity limit of 600,000 feet per week, regardless of the thickness.

What are the decision variables and constraints for this problem? The decision variables require some thought. There is only one way of producing  $\frac{1}{4}$ " material, three ways of producing  $\frac{3}{8}$ ", and two ways of producing  $\frac{1}{2}$ ". Thus, you will want to have at least the following decision variables. For numerical convenience, we measure length in thousands of feet:

$$B_{34} = 1,000\text{'s of feet of } \frac{1}{4}\text{" produced on } B_3,$$

$$B_{38} = 1,000\text{'s of feet of } \frac{3}{8}\text{" produced on } B_3,$$

$$B_{48} = 1,000\text{'s of feet of } \frac{3}{8}\text{" produced on } B_4,$$

$$B_{58} = 1,000\text{'s of feet of } \frac{3}{8}\text{" produced on } B_5,$$

$$B_{42} = 1,000\text{'s of feet of } \frac{1}{2}\text{" produced on } B_4,$$

$$B_{52} = 1,000\text{'s of feet of } \frac{1}{2}\text{" produced on } B_5.$$

For the objective function, we must have the profit contribution including labor costs. When this is done, we obtain:

Variable	Profit Contribution per Foot
$B_{34}$	0.01589
$B_{38}$	0.01789
$B_{48}$	0.01650
$B_{58}$	0.01522
$B_{42}$	0.01750
$B_{52}$	0.01622

Clearly, there will be four constraints corresponding to AMFC's three scarce machine resources and its shipping department capacity. There should be three more constraints due to the production requirements in the three products. For the machine capacity constraints, we want the number of hours required for 1,000 feet processed. For machine  $B_3$ , this figure is  $1,000/(60 \text{ min./hr.}) \times (150 \text{ ft./min.}) = 0.111111$  hours per 1,000 ft. Similar figures for  $B_4$  and  $B_5$  are 0.16667 hours per 1,000 ft. and 0.22222 hours per 1,000 feet.

The formulation can now be written:

$$\text{Maximize} = 15.89B_{34} + 17.89B_{38} + 16.5B_{48} + 15.22B_{58} + 17.5B_{42} + 16.22B_{52}$$

subject to

$$0.111111B_{34} + 0.111111B_{38} \leq 35 \quad \text{Machine}$$

$$0.16667B_{48} + 0.16667B_{42} \leq 35 \quad \text{capacities}$$

$$0.22222B_{58} + 0.22222B_{52} \leq 35 \quad \text{in hours}$$

$$B_{34} + B_{38} + B_{48} + B_{58} + B_{42} + B_{52} \leq 600 \quad \text{Shipping capacity in 1,000's of feet}$$

$$B_{34} \geq 218 \quad \text{Production}$$

$$B_{38} + B_{48} + B_{58} \geq 114 \quad \text{requirements}$$

$$B_{42} + B_{52} \geq 111 \quad \text{in 1,000's of feet}$$

Without the last three constraints, the problem is a simple product mix problem.

It is a worthwhile exercise to attempt to deduce the optimal solution just from cost arguments. The  $\frac{1}{4}$ " product can be produced on only machine  $B_3$ , so we know  $B_{34}$  is at least 218. The  $\frac{3}{8}$ " product is more profitable than the  $\frac{1}{4}$ " on machine  $B_3$ . Therefore, we can conclude that  $B_{34} = 218$  and  $B_{38}$  will take up the slack. The  $\frac{1}{2}$ " and the  $\frac{3}{8}$ " product can be produced on either  $B_4$  or  $B_5$ . In either case, the  $\frac{1}{2}$ " is more profitable per foot, so we know  $B_{48}$  and  $B_{58}$  will be no greater than absolutely necessary. The question is: What is "absolutely necessary"? The  $\frac{3}{8}$ " is more profitably run on  $B_3$  than on  $B_4$  or  $B_5$ . Therefore, it follows that we will satisfy the  $\frac{3}{8}$ " demand from  $B_3$  and, if sufficient, the remainder from  $B_4$  and then from  $B_5$ . Specifically, we proceed as follows:

$$\text{Set } B_{34} = 218.$$

This leaves a slack of  $35 - 218 \times 0.11111 = 10.78$  hours on  $B_3$ . This is sufficient to produce 97,000 feet of  $\frac{3}{8}$ ", so we conclude that:

$$B_{38} = 97.$$

The remainder of the  $\frac{3}{8}$ " demand must be made up from either machine  $B_4$  or  $B_5$ . It would appear that it should be done on machine  $B_4$  because the profit contribution for  $\frac{3}{8}$ " is higher on  $B_4$  than  $B_5$ . Note, however, that  $\frac{1}{2}$ " is also more profitable on  $B_4$  than  $B_5$  by exactly the same amount. Thus, we are indifferent. Let us arbitrarily use machine  $B_4$  to fill the rest of  $\frac{3}{8}$ " demand. Thus:

$$B_{48} = 17.$$

Now, any remaining capacity will be used to produce  $\frac{1}{2}$ " product. There are  $35 - 17 \times 0.16667 = 32.16667$  hours of capacity on  $B_4$ . At this point, we should worry about shipping capacity. We still have capacity for  $600 - 218 - 97 - 17 = 268$  in 1,000's of feet.  $B_{42}$  is more profitable than  $B_{52}$ , so we will make it as large as possible. Namely,  $32.16667/0.16667 = 193$ , so:

$$B_{42} = 193.$$

The remaining shipping capacity is  $268 - 193 = 75$ , so:

$$B_{52} = 75.$$

Any LP is in theory solvable by similar manual economic arguments, but the calculations could be very tedious and prone to errors of both arithmetic and logic. If we take the lazy route and solve it with LINGO, we get the same solution as our manual one:

```

Optimal solution found at step:          2
Objective value:                        10073.85
Variable      Value      Reduced Cost
  B34         218.00000      0.000000
  B38          97.00315      0.000000
  B48          16.99685      0.000000
  B58           0.00000      0.000000
  B42         192.99900      0.000000
  B52          75.00105      0.000000

Row    Slack or Surplus      Dual Price
  1          10073.85          1.000000
  2           0.000000         24.030240
  3           0.000000          7.679846
  4          18.333270          0.000000
  5           0.000000         16.220000
  6           0.000000         -3.000000
  7           0.000000         -1.000000
  8          157.000000          0.000000

```

Ranges in which the basis is unchanged:

```

Objective Coefficient Ranges
Variable      Current      Allowable      Allowable
Coefficient      Increase      Decrease
  B34         15.89000      3.000000      INFINITY
  B38         17.89000      INFINITY      2.670000
  B48         16.50000      1.000000      0.0
  B58         15.22000      0.000000      INFINITY
  B42         17.50000      0.0           1.000000
  B52         16.22000      1.280000      0.0

Right-hand Side Ranges
Row    Current      Allowable      Allowable
      RHS      Increase      Decrease
  2          35.00000      1.888520      9.166634
  3          35.00000      12.50043      13.75036
  4          35.00000      INFINITY      18.33327
  5          600.0000      82.50053      75.00105
  6          218.0000      97.00315      16.99685
  7          114.0000      157.0000      16.99685
  8          111.0000      157.0000      INFINITY

```

Notice *B58* is zero, but its reduced cost is also zero. This means *B58* could be increased (and *B48* decreased) without affecting profits. This is consistent with our earlier statement that we were indifferent between using *B48* and *B58* to satisfy the  $\frac{3}{8}$ " demand.



Below is a sets version of the problem:

```

!This is a sets version of the previous example;
MODEL:
SETS:
    MACHINE / B3, B4, B5 / : HPERWK, TIME;
!This is the coefficient for the time per day constraint;
    THICKNESS / FOURTH, EIGHT, HALF / : NEED;
!This is the amount of each thickness needed
to be produced;
    METHOD ( MACHINE, THICKNESS ) : VOLUME, PROFIT, POSSIBLE;
!VOLUME is the variable, PROFIT the objective coefficients, and POSSIBLE
is a Boolean representing whether it is possible to produce the given
thickness;
ENDSETS
DATA:
! Hours/week available on each machine;
    HPERWK = 35, 35, 35;
! Hours per 1000 feet for each machine;
    TIME = .11111 .16667 .22222;
! Amount needed of each product;
    NEED = 218 114 111;
! Profit by product and machine;
    PROFIT = 15.89, 17.89, 0,
             0, 16.5, 17.5,
             0, 15.22, 16.22;
! Which products can be made on which machine;
    POSSIBLE = 1, 1, 0,
              0, 1, 1,
              0, 1, 1;
! Shipping capacity per day;
    SHPERDAY = 600;
ENDDATA
!-----;
!Objective function;
MAX = @SUM( METHOD(I,J) : VOLUME(I,J) * PROFIT(I,J) );
@SUM( METHOD( K, L) : VOLUME( K, L) ) <= SHPERDAY;
!This is the max amount that can be made each day;
@FOR( MACHINE( N) :
    ! Maximum time each machine can be used/week.;
    @SUM( THICKNESS( M) :
        POSSIBLE(N,M) * VOLUME(N,M) * TIME(N) ) <= HPERWK(N) );
@FOR( THICKNESS( Q) :
    !Must meet demand for each thickness;
    @SUM( MACHINE(P) : POSSIBLE(P,Q) * VOLUME(P,Q) ) >= NEED(Q) );
END

```

## 6.4 Problems

1. Consider a manufacturer that produces two products, Widgets and Frisbees. Each product is made from the two raw materials, polyester and polypropylene. The following table gives the amounts required of each of the two products:

<b>Widgets</b>	<b>Frisbees</b>	<b>Raw Material</b>
3	5	Polyester
6	2	Polypropylene

Because of import quotas, the company is able to obtain only 12 units and 10 units of polyester and polypropylene, respectively, this month. The company is interested in planning its production for the next month. For this purpose, it is important to know the profit contribution of each product. These contributions have been found to be \$3 and \$4 for Widgets and Frisbees, respectively. What should be the amounts of Widgets and Frisbees produced next month?

2. The Otto Maddick Machine Tool Company produces two products, muffler bearings and torque amplifiers. One muffler bearing requires  $\frac{1}{8}$  hour of assembly labor, 0.25 hours in the stamping department, and 9 square feet of sheet steel. Each torque amplifier requires  $\frac{1}{3}$  hour in both assembly and stamping and uses 6 square feet of sheet steel. Current weekly capacities in the two departments are 400 hours of assembly labor and 350 hours of stamping capacity. Sheet steel costs 15 cents per square foot. Muffler bearings can be sold for \$8 each. Torque amplifiers can be sold for \$7 each. Unused capacity in either department cannot be laid off or otherwise fruitfully used.
- Formulate the LP useful in maximizing the weekly profit contribution.
  - It has just been discovered that two important considerations were not included.
    - Up to 100 hours of overtime assembly labor can be scheduled at a cost of \$5 per hour.
    - The sheet metal supplier only charges 12 cents per square foot for weekly usage in excess of 5000 square feet.

Which of the above considerations could easily be incorporated in the LP model and how? If one or both cannot be easily incorporated, indicate how you might nevertheless solve the problem.

3. Review the solution to the 5-product, 3-machine product mix problem introduced at the beginning of the chapter.
- What is the marginal value of an additional hour of capacity on each of the machines?
  - The current selling price of product *A* is \$5. What would the price have to be before we would produce any *A*?
  - It would be profitable to sell more of product *E* at \$4 if you could, but it is not profitable to sell *E* at \$3 per unit even though you can. What is the breakeven price at which you would be indifferent about selling any more *E*?
  - It is possible to gain additional capacity by renting by the hour automatic versions of each of the three machines. That is, they require no labor. What is the maximum hourly rate you would be willing to pay to rent each of the three types of automatic machines?

4. The Aviston Electronics Company manufactures motors for toys and small appliances. The marketing department is predicting sales of 6,100 units of the Dynamonster motor in the next quarter. This is a new high and meeting this demand will test Aviston's production capacities. A Dynamonster is assembled from three components: a shaft, base, and cage. It is clear that some of these components will have to be purchased from outside suppliers because of limited in-house capacity. The variable in-house production cost per unit is compared with the outside purchase cost in the following table.

<b>Component</b>	<b>Outside Cost</b>	<b>Inside Cost</b>
Shaft	1.21	0.81
Base	2.50	2.30
Cage	1.95	1.45

Aviston's plant consists of three departments. The time requirements in hours of each component in each department if manufactured in-house are summarized in the following table. The hours available for Dynamonster production are listed in the last row.

<b>Component</b>	<b>Cutting Department</b>	<b>Shaping Department</b>	<b>Fabrication Department</b>
Shaft	0.04	0.06	0.04
Base	0.08	0.02	0.05
Cage	0.07	0.09	0.06
Capacity	820	820	820

- What are the decision variables?
  - Formulate the appropriate LP.
  - How many units of each component should be purchased outside?
5. Buster Sod's younger brother, Marky Dee, operates three ranches in Texas. The acreage and irrigation water available for the three farms are shown below:

<b>Farm</b>	<b>Acreage</b>	<b>Water Available (acre feet)</b>
1	400	1500
2	600	2000
3	300	900

Three crops can be grown. However, the maximum acreage that can be grown of each crop is limited by the amount of appropriate harvesting equipment available. The three crops are described below:

<b>Crop</b>	<b>Total Harvesting Capacity (in acres)</b>	<b>Water Requirements (in acre-feet/acre)</b>	<b>Expected Profit (in \$/acre)</b>
Milo	700	6	400
Cotton	800	4	300
Wheat	300	2	100

Any combination of crops may be grown on a farm.

- a) What are the decision variables?
  - b) Formulate the LP.
6. Review the formulation and solution of the American Metal Fabricating process selection/product mix problem in this chapter. Based on the solution report:
- a) What is the value of an additional hour of capacity on the  $B_4$  machine?
  - b) What is the value of an additional 2 hours of capacity on the  $B_3$  machine?
  - c) By how much would one have to raise the profit contribution/1,000 ft. of  $\frac{1}{4}$ " material before it would be worth producing more of it?
  - d) If the speed of machine  $B_5$  could be doubled without changing the labor cost, what would it be worth per week? (Note labor on  $B_5$  is \$17/hour.)
7. A coupon recently appeared in an advertisement in the weekend edition of a newspaper. The coupon provided \$1 off the price of any size jar of Ocean Spray cranberry juice. The cost of the weekend paper was more than \$1.

Upon checking at a local store, we found two sizes available as follows:

<b>Size in oz.</b>	<b>Price</b>	<b>Price/oz. w/o Coupon</b>	<b>Price/oz. with Coupon</b>
32	2.09	.0653125	.0340625
48	2.89	.0602083	.039375

What questions, if any, should we ask in deciding which size to purchase? What should be our overall objective in analyzing a purchasing decision such as this?